

Universidade Federal de Santa Maria

Centro de Tecnologia

Desenvolvimento de Ontologia para o Banhado do Taim

Trabalho da Disciplina de **Inteligência Artificial**

Prof. Luis Alvaro

Autor:

Raul Hohgraefe da Cruz

Santa Maria, RS

2026

Sumário

1	Introdução	2
2	Modelagem da Ontologia	2
2.1	Visão geral	2
2.2	Hierarquia de classes	3
2.3	Propriedades de objeto	6
2.4	Propriedades de dados	7
2.5	Restrições	8
2.6	Decisões de modelagem	9
2.6.1	Decisão 1: EventoAtropelamento como classe central	9
2.6.2	Decisão 2: Separação entre CondicaoClimatica e NivelLuminosidade	9
2.6.3	Decisão 3: TrechoRodovia para especificar melhor o evento	9
2.7	Alternativas de modelagem consideradas e descartadas	9
2.7.1	Alternativa 1: condições climáticas como literais de dados	9
2.7.2	Alternativa 2: modelagem flat sem hierarquia de habitat	10
3	Povoamento da Ontologia	10
3.1	Estratégia de geração de dados	10
3.2	Exemplos de instâncias	12
3.3	Limitações dos dados	13
4	Consultas SPARQL	14
4.1	Consultas simples	14
4.2	Consultas com múltiplas relações	16
4.3	Consultas com filtros	19
4.4	Consultas com agregação	22
4.5	Consultas de cenários relevantes	24
5	Conclusão	27

1. Introdução

O Banhado do Taim é uma das áreas úmidas mais relevantes do Rio Grande do Sul, reconhecida como reserva ecológica federal e integrante do bioma Pampa. A região abriga uma biodiversidade expressiva, com espécies como capivaras (*Hydrochoerus hydrochaeris*), lontras (*Lontra longicaudis*), garças e diversas espécies de répteis, muitas das quais dependem diretamente dos ambientes alagados para sobrevivência e reprodução.

A BR-471, que atravessa a região, representa um dos principais fatores de pressão sobre essa fauna. Atropelamentos de animais silvestres constituem um problema recorrente no trecho que margeia o banhado, com registros distribuídos ao longo de dezenas de quilômetros e em variadas condições ambientais. Apesar disso, as informações sobre esses eventos tendem a permanecer dispersas em relatórios isolados, planilhas e bases de dados heterogêneas, dificultando análises mais abrangentes.

Este trabalho propõe a construção de uma ontologia OWL para o domínio ecológico e operacional do Banhado do Taim, com foco em eventos de atropelamento de fauna. O objetivo é estruturar formalmente o conhecimento sobre animais, habitats, rodovias, condições ambientais e fatores de risco, criando uma base de dados semântica consultável via SPARQL. A representação escolhida permite não apenas armazenar fatos, mas também expressar relações entre eles e potencialmente derivar inferências automáticas.

A ontologia foi desenvolvida em Python com a biblioteca `owlready2`, que permite definir classes, propriedades e restrições de forma programática e exportar o resultado em formato RDF/XML (`.owl`), compatível com ferramentas como o Protégé. O povoamento foi realizado com base nos dados reais da dissertação de mestrado de Nauderer (2014), que monitorou atropelamentos de fauna na BR-471 na região da Estação Ecológica do Taim, complementado com dados sintéticos plausíveis para as demais espécies. As consultas foram implementadas com a biblioteca `rdflib`.

2. Modelagem da Ontologia

2.1. Visão geral

A ontologia está identificada pela IRI <http://www.exemplo.org/taim#>, que serve como namespace único para todas as classes, propriedades e indivíduos. A escolha de uma URL como identificador segue o padrão do W3C para ontologias OWL: embora não precise ser um endereço acessível, garante a unicidade global dos termos.

A estrutura resultante compreende **52 classes**, **12 propriedades de objeto** e **17 propriedades de dados**, distribuídas em sete grupos conceituais que cobrem os requisitos mínimos do enunciado. Em relação à versão inicial, foram acrescentadas 9 classes de espécies para enriquecer a diversidade faunística da ontologia, todas com ocorrência confirmada na região do Taim.

2.2. Hierarquia de classes

Tabela 1: Hierarquia de classes da ontologia

Classe	Superclasse	Descrição
<i>Grupo: Animal, classes originais</i>		
Animal	,	Classe raiz para todos os animais
Mamifero	Animal	Mamíferos em geral
Ave	Animal	Aves em geral
Reptil	Animal	Répteis em geral
Capivara	Mamifero	<i>Hydrochoerus hydrochaeris</i>
Lontra	Mamifero	<i>Lontra longicaudis</i>
Tatu	Mamifero	<i>Dasypus novemcinctus</i>
Veado	Mamifero	<i>Ozotoceros bezoarticus</i>
GatoDomato	Mamifero	<i>Leopardus geoffroyi</i> , real (Nauderer 2014)
Graxaim	Mamifero	<i>Lycalopex gymnocercus</i>
Mao_pelada	Mamifero	<i>Procyon cancrivorus</i>
Preá	Mamifero	<i>Cavia aperea</i>
Tuco_tuco	Mamifero	<i>Ctenomys minutus</i> , endêmico
AveAquatica	Ave	Garças e afins
AveTerrestre	Ave	Quero-quero e afins
Serpente	Reptil	Serpentes da região
Tartaruga	Reptil	Tartarugas aquáticas
Lagarto	Reptil	<i>Tupinambis merianae</i>
Chimango	AveTerrestre	<i>Milvago chimango</i>
Carcara	AveTerrestre	<i>Caracara plancus</i>
Perdiz	AveTerrestre	<i>Rhynchotus rufescens</i>
<i>Grupo: Habitat</i>		
Habitat	,	Classe raiz para habitats
Banhado	Habitat	Áreas alagadas e banhados
CorpoDagua	Habitat	Corpos d'água em geral
VegetacaoRiparia	Habitat	Vegetação marginal
CampoAlagado	Habitat	Campos sujeitos a alagamento
Lagoa	CorpoDagua	Lagoas naturais
Rio	CorpoDagua	Rios e arroios
Vala	CorpoDagua	Valas de drenagem e irrigação

Continua na próxima página. . .

Continuação da Tabela 1

Classe	Superclasse	Descrição
<i>Grupo: Infraestrutura</i>		
Infraestrutura	,	Classe raiz para infraestrutura
Rodovia	Infraestrutura	Rodovias como a BR-471
TrechoRodovia	Infraestrutura	Segmentos quilométricos
Ponte	Infraestrutura	Pontes sobre corpos d'água
Acostamento	Infraestrutura	Acostamentos de rodovias
<i>Grupo: Evento e Condição</i>		
EventoAtropelamento	,	Registro de atropelamento
CondicaoAmbiental	,	Condições no momento do evento
CondicaoClimatica	CondicaoAmbiental	Chuva, neblina, céu aberto
NivelLuminosidade	CondicaoAmbiental	Diurno, noturno, crepuscular
Chuva	CondicaoClimatica	Precipitação
Neblina	CondicaoClimatica	Neblina
CeuAberto	CondicaoClimatica	Tempo limpo
Diurno	NivelLuminosidade	Período diurno
Noturno	NivelLuminosidade	Período noturno
Crepuscular	NivelLuminosidade	Amanhecer e entardecer
<i>Grupo: Risco e Tempo</i>		
FatorRisco	,	Fatores que aumentam o risco
TrafegoVeicular	FatorRisco	Volume de tráfego
VisibilidadeReduzida	FatorRisco	Visibilidade comprometida
ProximidadeCorpoHidrico	FatorRisco	Proximidade com água
AusenciaDePassagem	FatorRisco	Falta de passagem de fauna
PeriodoTemporal	,	Períodos de tempo
Estacao	PeriodoTemporal	Estações do ano
Horario	PeriodoTemporal	Faixas horárias

As Figuras 1 e 2 ilustram as hierarquias das classes **Animal** e **Habitat** conforme visualizadas no Protégé.




Figura 1: Hierarquia da classe **Animal** no Protégé, com subclasses **Mamifero**, **Ave** e **Reptil**




Figura 2: Hierarquia da classe **Habitat** no Protégé, com subclasses **Banhado**, **CorpoDagua** e derivados

2.3. Propriedades de objeto

As propriedades de objeto expressam relações entre indivíduos. A Tabela 2 lista as 12 propriedades definidas, com domínio, alcance e o tipo de relação que representam.

Tabela 2: Propriedades de objeto

Propriedade	Domínio	Alcance	Tipo
envolveAnimal	EventoAtropelamento	Animal	,
ocorreEm	EventoAtropelamento	TrechoRodovia	,
ocorreSob	EventoAtropelamento	CondicaoAmbiental	,
ocorreDurante	EventoAtropelamento	PeriodoTemporal	Temporal
possuiFatorRisco	EventoAtropelamento	FatorRisco	,
proximoA	TrechoRodovia	Habitat	Espacial
pertenceARodovia	TrechoRodovia	Rodovia	,
viveEm	Animal	Habitat	,
atravessaRodovia	Animal	Rodovia	,
contem	Habitat	Animal	,
afetaVisibilidade	CondicaoClimatica	VisibilidadeReduzida	,
aumentaRisco	CondicaoAmbiental	FatorRisco	,

A propriedade **ocorreDurante** representa a relação temporal obrigatória, vinculando cada evento ao seu período do dia (noturno, diurno ou crepuscular). A propriedade **proximoA** representa a relação espacial, conectando trechos da rodovia aos habitats adjacentes.

A Figura 3 apresenta o grafo gerado pelo plugin OntoGraf do Protégé, mostrando **EventoAtropelamento** no centro e suas conexões com as demais classes do domínio.




Figura 3: Grafo OntoGraf: EventoAtropelamento e suas relações com Animal, TrechoRodovia, CondicaoAmbienta, FatorRisco e PeriodoTemporal

2.4. Propriedades de dados

As propriedades de dados armazenam valores literais dos indivíduos. A Tabela 3 apresenta as 17 propriedades definidas.

Tabela 3: Propriedades de dados

Propriedade	Domínio	Tipo	Observação
nomeEspecie	Animal	string	Nome científico
nomePopular	Animal	string	Nome popular
kmInicial	TrechoRodovia	float	Km inicial do trecho
kmFinal	TrechoRodovia	float	Km final do trecho
coordenadaLat	TrechoRodovia	float	Espacial
coordenadaLon	TrechoRodovia	float	Espacial
nomeRodovia	Rodovia	string	Nome da rodovia

Continua na próxima página...

Continuação da Tabela 3

Propriedade	Domínio	Tipo	Observação
dataEvento	EventoAtropelamento	string	ISO (AAAA-MM-DD) Temporal
horarioEvento	EventoAtropelamento	string	HH:MM Temporal
resultadoFatal	EventoAtropelamento	boolean	Se o animal morreu
descricaoClima	CondicaoClimatica	string	Rótulo descritivo
temperatura	CondicaoClimatica	float	Em graus Celsius
pluviosidade	CondicaoClimatica	float	Em milímetros
nomeHabitat	Habitat	string	Nome do habitat
areaHabitat	Habitat	float	Em hectares
nivelRisco	FatorRisco	int	Escala de 1 a 5
descricaoRisco	FatorRisco	string	Descrição textual

2.5. Restrições

Quatro restrições de cardinalidade e existência foram definidas para garantir a coerência da ontologia:

- **EventoAtropelamento** deve envolver **pelo menos um** **Animal** (restrição *min* 1 na propriedade **envolveAnimal**).
- **EventoAtropelamento** deve ocorrer em **exatamente um** **TrechoRodovia** (restrição *exactly* 1 na propriedade **ocorreEm**).
- **TrechoRodovia** deve pertencer a **exatamente uma** **Rodovia** (restrição *exactly* 1 na propriedade **pertenceARodovia**).
- **Capivara** deve viver em **algum** **Banhado** ou **CorpoDagua** (restrição *some* na propriedade **viveEm**).

2.6. Decisões de modelagem

2.6.1. Decisão 1: *EventoAtropelamento* como classe central

Primeiramente, o *EventoAtropelamento* foi criado como uma propriedade entre *Animal* e *TrechoRodovia*, porém, como o atropelamento é o evento central e o motivo pelo qual estamos construindo essa ontologia, faz muito mais sentido que ele seja uma classe independente com seus próprios atributos. Essa foi uma decisão correta pois assim conseguimos detalhar mais esse evento e, como podemos ver na Figura 3, ele se torna o centro que liga todas as demais classes.

2.6.2. Decisão 2: *Separação entre CondicaoClimatica e NivelLuminosidade*

Embora ambas sejam *CondicaoAmbiental*, optou-se por tratá-las em subclasses distintas. Chuva e neblina são fenômenos meteorológicos independentes do horário, enquanto luminosidade não necessariamente, claro que haverá um impacto, porém ter essas duas informações em separado traz mais riqueza para o domínio e evita ambiguidade nas consultas que filtram apenas por condição climática.

2.6.3. Decisão 3: *TrechoRodovia* para especificar melhor o evento

Em vez de associar eventos diretamente à rodovia, criou-se a classe *TrechoRodovia* com quilometragem inicial e final e coordenadas geográficas. Dessa forma, conseguimos ter uma informação mais específica e com mais significado. Uma melhoria que pode ser óbvia de ver pelos dados em um trecho isolado seria algo que só conseguimos observar tendo essa classe em separado. Na prática, isso ficou evidente nos resultados da consulta C23: o trecho do km 541 concentrou 37 eventos, bem acima dos demais, o que é exatamente o hotspot principal do Setor 2 identificado por Nauderer (2014) — informação que se perderia se os eventos fossem associados apenas à BR-471 como um todo.

2.7. Alternativas de modelagem consideradas e descartadas

2.7.1. Alternativa 1: *condições climáticas como literais de dados*

Inicialmente, a condição climática era armazenada diretamente no evento como uma propriedade de dado — por exemplo, *descricaoClima* = "chuva_forte" dentro do próprio *EventoAtropelamento*. Parecia uma solução mais simples e direta. O problema apareceu na hora de escrever as consultas: para buscar todos os eventos sob algum tipo de chuva, seria necessário um *FILTER* com *CONTAINS* na string, o que é frágil e não aproveita nada da semântica da ontologia. Ao criar *Chuva* como subclasse de *CondicaoClimatica*, a consulta C12 consegue buscar *todos os eventos sob chuva*

de qualquer intensidade com uma linha só (`?condicao rdf:type ex:Chuva`), sem se preocupar com a descrição textual de cada instância.

2.7.2. Alternativa 2: modelagem flat sem hierarquia de habitat

Outra opção considerada foi tratar todos os habitats como instâncias de uma única classe `Habitat`, diferenciados apenas por um atributo `tipoHabitat` = "banhado", "lagoa", etc. Funcionaria para buscas simples, mas na prática a hierarquia se mostrou necessária. A consulta C11 — que recupera capivaras atropeladas em trechos próximos a banhados — usa `?habitat rdf:type ex:Banhado` diretamente. Com a modelagem flat, essa consulta exigiria um filtro por string no atributo de tipo, o que além de mais verboso, seria quebrado se alguém escrevesse "Banhado" com maiúscula em vez de "banhado". A hierarquia deixa isso mais robusto e, futuramente, permite que um raciocinador OWL infira automaticamente que uma lagoa é um corpo d'água sem precisar declarar isso explicitamente em cada instância.

3. Povoamento da Ontologia

3.1. Estratégia de geração de dados

A abordagem tem como fonte primária a dissertação de mestrado de Nauderer (2014), que monitorou atropelamentos de capivaras na BR-471 entre os km 536 e km 553, na área da Estação Ecológica do Taim, entre abril de 2010 e março de 2013. O estudo registrou 629 capivaras atropeladas ao longo de 132 dias de monitoramento, com taxa média de 0,278 ind./km/dia.

A estratégia combinou duas fontes complementares:

- **Dados reais (Nauderer 2014):** 50 eventos de capivara, 3 eventos de lontra e 3 eventos de gato-do-mato, totalizando 56 eventos com base empírica direta. Esses eventos representam aproximadamente **38% do total** de eventos da ontologia.
- **Dados sintéticos plausíveis:** 90 eventos distribuídos entre 14 espécies adicionais confirmadas na região, gerados com distribuições baseada nos padrões identificados no artigo (sazonalidade, período do dia, preferência por setor), sendo a base o GBIF.

A Tabela 4 resume os indivíduos criados.

Tabela 4: Resumo do povoamento

Tipo	Quantidade	Origem
Rodovia (BR-471)	1	Manual
Trechos (km 536 ao km 553)	14	Coordenadas reais
Habitats	8	Nomes reais da região
Condições climáticas	8	Sintético
Fatores de risco	5	Nauderer (2014)
Períodos temporais	7	Sintético
Animais reais	56	Nauderer (2014)
Animais sintéticos	56	Literatura regional
Eventos reais (capivara)	50	Nauderer (2014)
Eventos reais (lontra + gato)	6	Nauderer (2014)
Eventos sintéticos	90	Sintético (GBIF + Nauderer)
Total	301	

Os principais parâmetros extraídos do artigo e aplicados no script de povoamento foram:

- **Trechos e setores reais:** os 14 segmentos cobrem exatamente o trecho estudado (km 536–553), divididos nos três setores identificados por Nauderer (2014): Setor 1 (com tela SPF, km 536–539,4), Setor 2 (sem tela SPF, km 539,4–544,9, maior concentração de eventos) e Setor 3 (com tela SPF bem conservada, km 544,9–553). As zonas de transição nos km 540 e km 545 foram modeladas como trechos críticos separados, pois o estudo identificou esses pontos como hotspots associados à descontinuidade das telas.
- **Hotspots reais:** os pesos de distribuição dos eventos por trecho foram calibrados conforme os pontos de agregação identificados no estudo. O km 541 (Setor 2) recebeu o maior peso por concentrar o hotspot principal; os km 537, 538 e 539 formam o segundo cluster, e o km 546 representa o único hotspot do Setor 3.
- **Distribuição sazonal real:** Nauderer (2014) identificou que os meses de inverno (julho, agosto e setembro) concentram as maiores taxas de atropelamento, associadas à menor duração do dia somada à procura de comida das capivaras nesse mesmo período. Os meses de verão (janeiro a março) apresentam as menores taxas, mesmo com aumento do tráfego pelo turismo, fenômeno que o estudo atribui ao maior tempo de luz diurna. Essa distribuição foi reproduzida nos pesos mensais do script e aplicada também aos eventos sintéticos.

- **Distribuição por período do dia:** predominância noturna nos mamíferos (55–60% dos eventos no inverno), com bias diurno para aves e lagartos, refletindo os padrões de atividade conhecidos de cada grupo.
- **Taxa de fatalidade:** Todos os eventos reais foram marcados como fatais. Registros de campo em monitoramentos de atropelamentos correspondem predominantemente a animais encontrados mortos na pista; eventos de sobrevivência são raros e sistematicamente subnotificados [7]. Para os eventos sintéticos, adotou-se 80%.
- **Outras espécies reais:** Nauderer (2014) registrou, nas considerações finais, 3 lontras (*Lontra longicaudis*) e 3 gatos-do-mato-grande (*Leopardus geoffroyi*, sendo um melânico), todos no Setor 2, o único trecho sem cercas de proteção. Esses animais foram incluídos como dados reais, com a adição da classe `GatoDomato` na hierarquia de `Mamifero`.
- **Espécies sintéticas:** as 14 espécies adicionais foram selecionadas com base na literatura regional e registros do GBIF para a área do Taim. Para cada espécie foram definidos setor preferencial, habitat e padrão de atividade, de forma que os eventos sintéticos reflitam a ecologia de cada grupo, mão-pelada semi-aquática preferindo o Setor 2 próximo ao rio, lagartos com bias diurno, aves carniceiras (chimango, carcará) distribuídas ao longo de todos os setores, etc.

3.2. Exemplos de instâncias

A seguir, alguns exemplos de indivíduos criados na ontologia, em pseudonotação Turtle:

Listing 1: Exemplos de indivíduos na ontologia

```
ex:capivara_001  a  ex:Capivara  ;
    ex:nomeEspecie    "Hydrochoerus hydrochaeris"  ;
    ex:nomePopular    "Capivara"  ;
    ex:viveEm          ex:banhado_taim_central  .

ex:trecho_BR471_km541  a  ex:TrechoRodovia  ;
    ex:kmInicial        541.0  ;
    ex:kmFinal          542.0  ;
    ex:coordenadaLat    -33.5565  ;
    ex:coordenadaLon    -53.3913  ;
    ex:pertenceARodovia  ex:rodovia_BR471  ;
    ex:proximoA          ex:banhado_taim_central  .

ex:evento_001  a  ex:EventoAtropelamento  ;
    ex:envolveAnimal    ex:capivara_024  ;
    ex:ocorreEm          ex:trecho_BR471_km544  ;
    ex:ocorreSob         ex:clima_chuva_fraca  ;
    ex:ocorreDurante     ex:periodo_diurno  ;
```

```

ex:possuiFatorRisco ex:risco_ausencia_spf ;
ex:dataEvento      "2010-11-17" ;
ex:horarioEvento   "13:31" ;
ex:resultadoFatal  true .

ex:gato_domato_001 a ex:GatoDomato ;
ex:nomeEspecie     "Leopardus geoffroyi" ;
ex:nomePopular     "Gato-do-mato-grande (mel nico)" ;
ex:viveEm          ex:campo_alagado_sul .

```

A Figura 4 confirma o indivíduo `evento_001` no Protégé, com todas as propriedades de objeto e de dados devidamente preenchidas.



Figura 4: Indivíduo `evento_001` no Protégé com todas as propriedades preenchidas

3.3. Limitações dos dados

Algumas limitações merecem registro:

- **Representatividade das capivaras:** os 629 registros originais foram representados por 50 indivíduos e 50 eventos, por razões de escala. A distribuição estatística foi preservada, mas a granularidade individual é uma abstração.
- **Eventos sintéticos sem validação de campo:** os 90 eventos das 14 espécies adicionais seguem distribuições plausíveis baseadas na ecologia de cada grupo, mas não têm correspondência a registros reais de atropelamento para o Taim especificamente.
- **Ausência de dados de tráfego:** os fatores de risco relacionados a volume de tráfego veicular são atribuídos por setor, sem dados quantitativos reais do DNIT para os trechos modelados.
- **Dados climáticos:** as condições climáticas foram definidas manualmente como instâncias-tipo (chuva fraca, forte, neblina, céu aberto), sem vínculo a séries históricas do INMET para as datas específicas dos eventos.

4. Consultas SPARQL

As 30 consultas foram implementadas e executadas com a biblioteca `rdflib` sobre o arquivo `taim_povoado.owl`. Essa ferramenta foi adotada por oferecer suporte completo ao SPARQL 1.1, incluindo *property paths* (`rdfs:subClassOf*`) e funções de agregação — recursos não suportados integralmente pelo plugin SPARQL do Protégé. A ferramenta Protégé foi utilizada apenas para validação visual da estrutura da ontologia, navegação pelos indivíduos e geração dos grafos apresentados nas seções anteriores, pois não consegui realizar as consultas pelo Protégé em meu computador Linux.

Todas as consultas utilizam os seguintes prefixos:

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX ex: <http://www.exemplo.org/taim#>
```

As consultas estão organizadas em cinco categorias: simples [S], múltiplas relações [M], com filtros [F], agregação [A] e cenários relevantes [C].

4.1. Consultas simples

C01, Liste todos os animais com nome científico e popular

```
SELECT ?animal ?especie ?popular
WHERE {
  ?animal rdf:type ?tipo .
  ?tipo rdfs:subClassOf* ex:Animal .
  ?animal ex:nomeEspecie ?especie .
  ?animal ex:nomePopular ?popular .
}
ORDER BY ?popular
```

Resultado: 112 animais retornados, abrangendo todas as espécies reais e sintéticas da ontologia, capivaras, lontras, gatos-do-mato, graxaim, mão-pelada, quero-quero, serpentes, tartarugas, entre outros.

C02, Liste todos os trechos da BR-471 com quilômetros e coordenadas

```
SELECT ?trecho ?kmInicial ?kmFinal ?lat ?lon
WHERE {
  ?trecho rdf:type ex:TrechoRodovia .
  ?trecho ex:kmInicial ?kmInicial .
  ?trecho ex:kmFinal ?kmFinal .
  ?trecho ex:coordenadaLat ?lat .
  ?trecho ex:coordenadaLon ?lon .
```

```
}  
ORDER BY ?kmInicial
```

Resultado: 14 trechos do km 536 ao km 553, com coordenadas geográficas reais correspondentes aos três setores de Nauderer (2014).

C03, Liste todos os habitats com nome e área

```
SELECT ?habitat ?nome ?area  
WHERE {  
  ?habitat rdf:type ?tipo .  
  ?tipo    rdfs:subClassOf* ex:Habitat .  
  ?habitat ex:nomeHabitat ?nome .  
  ?habitat ex:areaHabitat ?area .  
}  
ORDER BY DESC(?area)
```

Resultado: 8 habitats, do Banhado do Taim Central (9.800 ha) à Vala de Drenagem BR-471 (180 ha).

C04, Liste as condições climáticas com temperatura e pluviosidade

```
SELECT ?clima ?descricao ?temperatura ?pluviosidade  
WHERE {  
  ?clima rdf:type ?tipo .  
  ?tipo  rdfs:subClassOf* ex:CondicaoClimatica .  
  ?clima ex:descricaoClima ?descricao .  
  ?clima ex:temperatura      ?temperatura .  
  ?clima ex:pluviosidade     ?pluviosidade .  
}  
ORDER BY ?descricao
```

Resultado: 8 condições climáticas com temperatura de 12,5°C (neblina densa) a 26,0°C (céu aberto diurno).

C05, Liste os fatores de risco por nível

```
SELECT ?fator ?descricao ?nivel  
WHERE {  
  ?fator rdf:type ?tipo .  
  ?tipo  rdfs:subClassOf* ex:FatorRisco .  
  ?fator ex:descricaoRisco ?descricao .  
  ?fator ex:nivelRisco      ?nivel .  
}  
ORDER BY DESC(?nivel)
```


Resultado: 5 fatores de risco diretamente embasados em Nauderer (2014), com nível máximo 5 para ausência de SPF no Setor 2.

C06, Liste todos os eventos de atropelamento com data, horário e fatalidade

```
SELECT ?evento ?data ?horario ?fatal
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:dataEvento     ?data .
  ?evento ex:horarioEvento  ?horario .
  ?evento ex:resultadoFatal ?fatal .
}
ORDER BY ?data ?horario
```

Resultado: 146 eventos no total, com datas entre 2010 e 2023. Os eventos de 2010 a 2013 correspondem aos dados reais de capivaras, lontras e gatos-do-mato; os demais são sintéticos.

4.2. Consultas com múltiplas relações

C07, Animais envolvidos em eventos e os trechos onde ocorreram

```
SELECT ?popular ?trecho ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreEm      ?trecho .
  ?evento ex:dataEvento    ?data .
  ?animal ex:nomePopular   ?popular .
}
ORDER BY ?popular
```

Resultado: 146 registros cruzando espécie, trecho e data, mostrando a diversidade faunística distribuída pelos 14 trechos.

C08, Eventos com animal, trecho, clima e período do dia

```
SELECT ?evento ?popular ?trecho ?clima ?periodo
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreEm      ?trecho .
  ?evento ex:ocorreSob    ?condicao .
  ?evento ex:ocorreDurante ?periodo .
  ?animal ex:nomePopular   ?popular .
  ?condicao ex:descricaoClima ?clima .
}
```

```
}

```

Resultado: 146 eventos com todas as dimensões contextuais preenchidas.

C09, Trechos da BR-471 e os habitats adjacentes a cada um

```
SELECT ?trecho ?kmInicial ?kmFinal ?nomeHabitat
WHERE {
  ?trecho rdf:type ex:TrechoRodovia .
  ?trecho ex:kmInicial ?kmInicial .
  ?trecho ex:kmFinal ?kmFinal .
  ?trecho ex:proximoA ?habitat .
  ?habitat ex:nomeHabitat ?nomeHabitat .
}
ORDER BY ?kmInicial

```

Resultado: 14 associações trecho-habitat, confirmando que todos os trechos estão adjacentes a banhados do Taim, relação espacial central do domínio.

C10, Eventos fatais com o animal envolvido e o fator de risco

```
SELECT ?popular ?especie ?fatorDesc ?trecho
WHERE {
  ?evento rdf:type ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreEm ?trecho .
  ?evento ex:possuiFatorRisco ?fator .
  ?evento ex:resultadoFatal "true"^^xsd:boolean .
  ?animal ex:nomePopular ?popular .
  ?animal ex:nomeEspecie ?especie .
  ?fator ex:descricaoRisco ?fatorDesc .
}

```

Resultado: 126 eventos fatais com identificação do fator de risco associado, com predomínio de “Ausência de SPF” e “Proximidade de corpo hídrico”.

C11, Capivaras atropeladas em trechos próximos a banhados (dados reais Nauderer 2014)

Esta consulta demonstra a cadeia de raciocínio central do domínio: animal → evento → trecho → habitat.

```
SELECT ?evento ?trecho ?nomeHabitat ?data
WHERE {
  ?evento rdf:type ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .

```

```

?evento ex:ocorreEm ?trecho .
?evento ex:dataEvento ?data .
?animal rdf:type ex:Capivara .
?trecho ex:proximoA ?habitat .
?habitat rdf:type ex:Banhado .
?habitat ex:nomeHabitat ?nomeHabitat .
}
ORDER BY ?data

```

Resultado: 50 eventos de capivara em trechos adjacentes a banhados, todos com datas entre 2010 e 2013, confirmando a cobertura completa dos dados reais do artigo.

C12, Animais atropelados sob chuva com trecho e horário

```

SELECT ?popular ?trecho ?horario ?data
WHERE {
  ?evento rdf:type ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreEm ?trecho .
  ?evento ex:ocorreSob ?condicao .
  ?evento ex:horarioEvento ?horario .
  ?evento ex:dataEvento ?data .
  ?animal ex:nomePopular ?popular .
  ?condicao rdf:type ex:Chuva .
}
ORDER BY ?data

```

Resultado: 45 eventos sob condição de chuva de qualquer intensidade, envolvendo 12 espécies diferentes.

C13, Eventos com todas as cinco relações simultâneas

```

SELECT ?popular ?trecho ?clima ?periodo ?fator
WHERE {
  ?evento rdf:type ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreEm ?trecho .
  ?evento ex:ocorreSob ?condicao .
  ?evento ex:ocorreDurante ?per .
  ?evento ex:possuiFatorRisco ?fat .
  ?animal ex:nomePopular ?popular .
  ?condicao ex:descricaoClima ?clima .
  ?fat ex:descricaoRisco ?fator .
  BIND(STRAFTER(STR(?per), "#") AS ?periodo)
}

```

Resultado: 146 eventos com todas as cinco dimensões contextuais preenchidas, confirmando a integridade do povoamento.

4.3. Consultas com filtros

C14, Eventos ocorridos no período noturno (20h–05h59)

```
SELECT ?evento ?popular ?horario ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:horarioEvento ?horario .
  ?evento ex:dataEvento    ?data .
  ?animal ex:nomePopular   ?popular .
  FILTER (
    xsd:integer(SUBSTR(?horario, 1, 2)) >= 20 ||
    xsd:integer(SUBSTR(?horario, 1, 2)) <= 5
  )
}
ORDER BY ?horario
```

Resultado: 67 eventos noturnos (46% do total), confirmando a predominância noturna dos atropelamentos, especialmente para mamíferos.

C15, Eventos com chuva forte (pluviosidade > 50 mm)

```
SELECT ?evento ?popular ?pluviosidade ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreSob     ?condicao .
  ?evento ex:dataEvento    ?data .
  ?animal ex:nomePopular   ?popular .
  ?condicao ex:pluviosidade ?pluviosidade .
  FILTER (?pluviosidade > 50)
}
ORDER BY DESC(?pluviosidade)
```

Resultado: 10 eventos sob chuva forte (55 mm), abrangendo capivaras, tuco-tuco, carcará, perdiz e garça-branca-grande.

C16, Eventos fatais nos meses de inverno (julho, agosto e setembro)

Esta consulta é diretamente derivada do padrão sazonal identificado por Nauderer (2014): o inverno concentra as maiores taxas de atropelamento.

```
SELECT ?evento ?popular ?data ?horario
```

```

WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?evento ex:resultadoFatal "true"^^xsd:boolean .
  ?evento ex:dataEvento     ?data .
  ?evento ex:horarioEvento  ?horario .
  ?animal ex:nomePopular    ?popular .
  FILTER (
    xsd:integer(SUBSTR(?data, 6, 2)) >= 7 &&
    xsd:integer(SUBSTR(?data, 6, 2)) <= 9
  )
}
ORDER BY ?data

```

Resultado: 58 eventos fatais nos meses de inverno, o maior agrupamento sazonal da base, validando a distribuição implementada no script.

C17, Eventos nos hotspots principais (km 537–542, Setores 1 e 2)

```

SELECT ?evento ?popular ?kmInicial ?kmFinal ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?evento ex:ocorreEm       ?trecho .
  ?evento ex:dataEvento     ?data .
  ?animal ex:nomePopular    ?popular .
  ?trecho ex:kmInicial      ?kmInicial .
  ?trecho ex:kmFinal        ?kmFinal .
  FILTER (?kmInicial >= 537 && ?kmFinal <= 542)
}
ORDER BY ?kmInicial

```

Resultado: 78 eventos concentrados nos hotspots dos Setores 1 e 2, confirmando que esses trechos reúnem a maior parte dos atropelamentos modelados.

C18, Eventos com temperatura abaixo de 15°C (condições de inverno)

```

SELECT ?evento ?popular ?temperatura ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?evento ex:ocorreSob      ?condicao .
  ?evento ex:dataEvento     ?data .
  ?animal ex:nomePopular    ?popular .
  ?condicao ex:temperatura    ?temperatura .
  FILTER (?temperatura < 15)
}

```

```
}
ORDER BY ?temperatura
```

Resultado: 48 eventos sob temperatura inferior a 15°C, abrangendo 12 espécies e com predomínio de mamíferos noturnos.

C19, Todos os mamíferos cadastrados na ontologia

```
SELECT DISTINCT ?animal ?especie ?popular
WHERE {
  ?animal rdf:type ?tipo .
  ?tipo    rdfs:subClassOf* ex:Mamifero .
  ?animal ex:nomeEspecie ?especie .
  ?animal ex:nomePopular ?popular .
}
ORDER BY ?popular
```

Resultado: 80 mamíferos, de 10 espécies diferentes, capivaras, lontras, gatos-do-mato, veados, tatus, graxaim, mão-pelada, preá e tuco-tuco.

C20, Eventos ocorridos no período crepuscular (5h–7h59 e 17h–19h59)

```
SELECT ?evento ?popular ?horario ?data
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:horarioEvento ?horario .
  ?evento ex:dataEvento    ?data .
  ?animal ex:nomePopular    ?popular .
  FILTER (
    (xsd:integer(SUBSTR(?horario, 1, 2)) >= 5 &&
     xsd:integer(SUBSTR(?horario, 1, 2)) <= 7)
    ||
    (xsd:integer(SUBSTR(?horario, 1, 2)) >= 17 &&
     xsd:integer(SUBSTR(?horario, 1, 2)) <= 19)
  )
}
ORDER BY ?horario
```

Resultado: 52 eventos crepusculares, período de transição que reúne tanto mamíferos quanto aves saindo para forrageamento.

C21, Mamíferos em habitats com área acima de 5.000 ha

```
SELECT ?popular ?especie ?nomeHabitat ?area
WHERE {
```

```

?evento  rdf:type          ex:EventoAtropelamento .
?evento  ex:envolveAnimal  ?animal .
?evento  ex:ocorreEm       ?trecho .
?animal  rdf:type          ?tipo .
?tipo    rdfs:subClassOf*  ex:Mamifero .
?animal  ex:nomePopular    ?popular .
?animal  ex:nomeEspecie    ?especie .
?trecho   ex:proximoA       ?habitat .
?habitat  ex:nomeHabitat    ?nomeHabitat .
?habitat  ex:areaHabitat    ?area .
FILTER (?area > 5000)
}

```

Resultado: 97 ocorrências de mamíferos em grandes áreas de banhado, mostrando que a maior parte dos atropelamentos de mamíferos acontece nas bordas dos grandes banhados.

4.4. Consultas com agregação

C22, Ranking de atropelamentos por espécie

```

SELECT ?popular (COUNT(?evento) AS ?total)
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?animal ex:nomePopular    ?popular .
}
GROUP BY ?popular
ORDER BY DESC(?total)

```

Tabela 5: Resultado da consulta C22

Espécie	Total
Capivara	50
Mão-pelada	10
Quero-quero	10
Tatu-galinha	7
Graxaim-do-campo	7
Preá	7
Chimango	7
Serpente-d'água	7
Veado-campeiro	5
Tuco-tuco	5

A capivara lidera com 50 eventos, o que reflete diretamente o uso de dados reais. A diferença em relação às demais espécies é proposital e coerente com o artigo de referência.

C23, Contagem de eventos por trecho da rodovia

```
SELECT ?trecho ?kmInicial (COUNT(?evento) AS ?total)
WHERE {
    ?evento rdf:type          ex:EventoAtropelamento .
    ?evento ex:ocorreEm      ?trecho .
    ?trecho ex:kmInicial ?kmInicial .
}
GROUP BY ?trecho ?kmInicial
ORDER BY DESC(?total)
```

Resultado: O trecho do km 541 lidera com 37 eventos, bem à frente dos demais. Esse é exatamente o hotspot principal do Setor 2 identificado por Nauderer (2014), confirmando que os pesos de distribuição foram reproduzidos corretamente.

C24, Distribuição de eventos por condição climática

```
SELECT ?descricao (COUNT(?evento) AS ?total)
WHERE {
    ?evento rdf:type          ex:EventoAtropelamento .
    ?evento ex:ocorreSob      ?condicao .
    ?condicao ex:descricaoClima ?descricao .
}
GROUP BY ?descricao
ORDER BY DESC(?total)
```

Resultado: Neblina leve e céu aberto noturno lideram com 25 eventos cada, seguidos por céu aberto diurno (21) e chuva fraca (20), refletindo a distribuição sazonal com maior presença de neblina no inverno.

C25, Proporção de eventos fatais vs. não fatais

```
SELECT ?fatal (COUNT(?evento) AS ?total)
WHERE {
    ?evento rdf:type          ex:EventoAtropelamento .
    ?evento ex:resultadoFatal ?fatal .
}
GROUP BY ?fatal
ORDER BY DESC(?total)
```


Resultado: 126 eventos fatais (86%) e 20 não fatais (14%). Os eventos reais são 100% fatais por definição; os sintéticos têm taxa de 80%, resultando nessa proporção geral.

C26, Estatísticas de temperatura nos eventos

```
SELECT
  (AVG(?t) AS ?tempMedia)
  (MIN(?t) AS ?tempMin)
  (MAX(?t) AS ?tempMax)
WHERE {
  ?evento    rdf:type          ex:EventoAtropelamento .
  ?evento    ex:ocorreSob     ?condicao .
  ?condicao   ex:temperatura  ?t .
}
```

Resultado: Média de 17,0°C, mínimo de 12,5°C e máximo de 26,0°C. A média abaixo de 18°C reflete o peso maior dos eventos de inverno na base.

4.5. Consultas de cenários relevantes

C27, Trechos sem SPF com fator de risco nível ≥ 4 próximos a banhado

Esta consulta identifica trechos que combinam dois fatores críticos: proximidade com habitat de banhado e fator de risco de nível 4 ou 5, ou seja, os pontos onde a ausência ou falha do SPF se sobrepõe à rota natural de fauna.

```
SELECT DISTINCT ?trecho ?kmInicial ?nomeHabitat ?fatorDesc ?nivel
WHERE {
  ?evento    rdf:type          ex:EventoAtropelamento .
  ?evento    ex:ocorreEm      ?trecho .
  ?evento    ex:possuiFatorRisco ?fator .
  ?trecho    ex:kmInicial     ?kmInicial .
  ?trecho    ex:proximoA      ?habitat .
  ?habitat   rdf:type          ex:Banhado .
  ?habitat   ex:nomeHabitat    ?nomeHabitat .
  ?fator     ex:descricaoRisco ?fatorDesc .
  ?fator     ex:nivelRisco     ?nivel .
  FILTER (?nivel >= 4)
}
ORDER BY DESC(?nivel) ?kmInicial
```

Resultado: 15 trechos distintos identificados como zonas críticas, todos com presença de eventos e adjacentes ao Banhado do Taim. Os trechos km 540 a km 545 (Setor 2 e transições) aparecem com fator nível 5, e os do Setor 1 com nível 4 pelo SPF danificado.

C28, Ranking de espécies por atropelamentos fatais

```

SELECT ?popular ?especie (COUNT(?evento) AS ?atropelamentosFatais)
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?evento ex:resultadoFatal "true"^^xsd:boolean .
  ?animal ex:nomePopular    ?popular .
  ?animal ex:nomeEspecie    ?especie .
}
GROUP BY ?popular ?especie
ORDER BY DESC(?atropelamentosFatais)

```

Resultado: A capivara lidera com 50 óbitos (dados reais), seguida por mão-pelada, preá e serpente-d'água com 7 cada. Lontra e gato-do-mato aparecem com 3 óbitos cada, refletindo exatamente os registros do artigo de Nauderer (2014).

C29, Gato-do-mato e lontra: perfil de atropelamento no Setor 2 (dados reais)

Esta consulta isola os dois grupos de espécies não-capivara com dados reais na ontologia, permitindo verificar se os atributos foram inseridos corretamente e examinar o perfil temporal desses eventos.

```

SELECT ?popular ?especie ?data ?horario ?kmInicial ?fatorDesc
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal  ?animal .
  ?evento ex:ocorreEm       ?trecho .
  ?evento ex:possuiFatorRisco ?fator .
  ?evento ex:dataEvento     ?data .
  ?evento ex:horarioEvento  ?horario .
  ?animal ex:nomePopular    ?popular .
  ?animal ex:nomeEspecie    ?especie .
  ?trecho ex:kmInicial      ?kmInicial .
  ?fator  ex:descricaoRisco  ?fatorDesc .
  FILTER (
    CONTAINS(LCASE(STR(?especie)), "lontra") ||
    CONTAINS(LCASE(STR(?especie)), "leopardus")
  )
}
ORDER BY ?data

```

Resultado: 6 eventos, 3 lontras e 3 gatos-do-mato, todos no Setor 2, todos com fator “Ausência de SPF” e com datas entre 2010 e 2011, correspondendo exatamente ao período e ao setor registrados por Nauderer (2014).

C30, Diagnóstico por período do dia: distribuição por grupo faunístico

Esta consulta agrega os eventos por grupo taxonômico (Mamífero, Ave, Reptil) e período do dia, permitindo verificar se os padrões de atividade de cada grupo foram reproduzidos corretamente no poblamento.

```
SELECT ?grupo ?periodo (COUNT(?evento) AS ?total)
WHERE {
  ?evento rdf:type          ex:EventoAtropelamento .
  ?evento ex:envolveAnimal ?animal .
  ?evento ex:ocorreDurante ?per .
  ?animal rdf:type          ?tipo .
  ?tipo    rdfs:subClassOf* ?grupoClass .
  FILTER (?grupoClass IN (ex:Mamifero, ex:Ave, ex:Reptil))
  BIND(STRATER(STR(?grupoClass), "#") AS ?grupo)
  BIND(STRATER(STR(?per),          "#") AS ?periodo)
}
GROUP BY ?grupo ?periodo
ORDER BY ?grupo DESC(?total)
```

Tabela 6: Resultado da consulta C30, distribuição por grupo e período

Grupo	Período	Total
Ave	periodo_diurno	19
Ave	periodo_crepuscular	11
Ave	periodo_noturno	2
Mamifero	periodo_noturno	53
Mamifero	periodo_crepuscular	26
Mamifero	periodo_diurno	18
Reptil	periodo_crepuscular	7
Reptil	periodo_noturno	6
Reptil	periodo_diurno	4

O resultado é exatamente o que se esperaria da ecologia desses grupos: mamíferos com forte predominância noturna (55% dos eventos do grupo), aves predominantemente diurnas (59%), e répteis com leve bias crepuscular. Isso confirma que os parâmetros de distribuição de horário implementados no script foram reproduzidos corretamente na base.

5. Conclusão

Este trabalho demonstrou a viabilidade de construir uma ontologia OWL funcional para o domínio de atropelamentos de fauna no Banhado do Taim, desde a modelagem até o povoamento e a consulta dos dados.

A escolha de **EventoAtropelamento** como classe central mostrou-se adequada, pois permite relacionar de forma estruturada todas as dimensões do fenômeno: o animal envolvido, o trecho da rodovia, a condição climática, o período do dia e o fator de risco. Isso torna cada registro muito mais rico do que uma simples tupla em uma planilha.

A reformulação do povoamento para usar os dados reais de Nauderer (2014) foi muito eficiente, pois deu motivos para validar nossa ontologia e dar uma direção mais real ao trabalho. Os eventos de capivara agora têm embasamento empírico direto, com datas, hotspots e distribuição sazonal reais, e os demais animais foram inseridos com parâmetros baseados no estudo.

As 30 consultas SPARQL cobriram desde buscas simples por classe até análises multidimensionais com agregação, e os resultados fazem sentido ecológico: o km 541 lidera em eventos, o inverno concentra os fatais, os mamíferos são majoritariamente noturnos e as aves predominantemente diurnas. A consulta C27 mostrou como combinar relações espaciais e fatores de risco para identificar os trechos prioritários para intervenção, e a C29 confirmou que os dados reais de lontra e gato-do-mato foram inseridos com fidelidade.

Como trabalho futuro, a integração com Machine Learning seria de bom tom, pois além de ajudar no estudo do conteúdo da disciplina de Inteligência Artificial, tornaria nossa ontologia ainda mais útil e proveitosa para o Banhado do Taim, adicionando funcionalidades de Predição e Diagnóstico.

Referências

- [1] Lamy, J.B. Owlready2: A Python library and infrastructure for ontology-oriented programming. *Artificial Intelligence in Medicine*, v. 117, 2021.
- [2] Nauderer, R. *Atropelamentos de capivaras (Hydrochoerus hydrochaeris) na BR-471, Estação Ecológica do Taim, RS*. Dissertação de Mestrado, Universidade Federal do Rio Grande, FURG, Rio Grande, RS, 2014.
- [3] Musen, M.A. The Protégé project: A look back and a look forward. *AI Matters*, v. 1, n. 4, p. 4–12, 2015.
- [4] ICMBio. Estação Ecológica do Taim. Disponível em: <https://www.gov.br/icmbio>. Acesso em: jun. 2026.

- [5] Ministério do Meio Ambiente. Projeto Atropelamentos de Fauna Silvestre. Disponível em: <https://www.gov.br/mma>. Acesso em: jun. 2026.
- [6] Bager, A.; Fontoura, V. Evaluation of the effectiveness of a wildlife roadkill mitigation system in wetland habitat. *Ecological Engineering*, v. 53, p. 31–38, 2013.
- [7] Coelho, I.P. et al. Anurans on a news highway crossing the Atlantic Forest in southern Brazil. *Journal of Herpetology*, v. 46, n. 2, p. 151–161, 2012.